RESPONSE DOCUMENT

INITIAL QUALIFICATION PROCEDURE –
EXTENSION TO THIRD LIST OF ISSUES –
SIMCYP SIMULATOR AND REQUEST FOR ADDITIONAL INFORMATION FROM SIMCYP
ITEMS 2-4

February 19, 2025

CONTACT: **KAREN ROWLAND YEO**
SENIOR VICE-PRESIDENT
CERTARA UK LTD
LEVEL 2-ACERO
1 CONCOURSE WAY
SHEFFIELD S1 2BJ
UK

EMAIL: KAREN.YEO@CERTARA.COM

Contents

Third List of Issues

Impact of inhibitor strength

Following the presentation of Certara's answers to the third list of issues, it was noted that the following **Next Step and Related Comment** needed to be addressed:

- EMA stated that Certara need to provide an updated analysis including a nonlinear function for inhibitor strength and an additional figure for goodness of fit inclusive of stratification of inhibitor strength.

Additional request: Develop and present the results of a new hierarchical model D with varying GMR variance and between study variability according to degree of inhibition.

Answer strategy

To answer these questions, model C (original script "EMA extended m202.stan", Figure 1) was modified to include three levels for inhibitor strength (*weak*, *moderate*, *strong*). Two versions of it were studied: Model D ("EMA extended m203.stan"), with a linear covariate model for strength, and Model E ("EMA extended m204.stan"), with a quadratic covariate model. Their equations are given in the following sections. Those models make full use of all the data and do not discard the strong inhibitor studies (an essential advantage over model C).



Figure 1: Directed acyclic graph representations of the model dependencies between variables and parameters in models D and E (based on model C). Literature data are in red; Simcyp® predictions in blue; Latent variables in grey; Estimands in black.

Data

We used the same full data file as previously (files Full DB 08.csv and Processed full DB 08.csv).

Bias and imprecision model D

This model is based on model C, extended to consider three inhibitor strengths. Strength is a linear covariate.

The GMR observed in study i (out of $N_{studies}$) is still assumed to be lognormally distributed around a true GMR ($logGMR_i$, in log-space), with true standard deviation $\sigma_{GMR,i}$ (in log-space):

$$GMR_{i,obs} \sim \mathcal{LN}\left(logGMR_i, \sigma_{GMR,i}\right) \quad (1)$$

Similar to EMA's implementation of the previous model, the true GMR is assumed equal to the Simcyp-predicted value, $GMR_{i,pred}$, but it is now corrected by a study-specific bias $\beta'_{GMR,i}$:

$$logGMR_i = \log\left(GMR_{i,pred}\right) + \beta'_{GMR,i} \quad (2)$$

Bias for study i depends on the CYP enzyme mediating the studied interaction, on the type of interaction occurring (competitive *vs*. mechanism-based), and on the strength of the inhibitor used in study i:

$$\beta'_{GMR,i} = \delta_{GMR,CYP_i} + \mathbb{I}_{inh_i} \times \delta_{GMR,inh} + \mathbb{I}_{str_i} \times \delta_{GMR,str} \quad (3)$$

Parameter δ_{GMR,CYP_i} measures the bias affecting simulations of the CYP examined in study i; $\mathbb{I}_{inh_i}$ is an indicator function taking the value 1 if study i examined a mechanism-based inhibition and value 0 if it examined a competitive inhibition; $\delta_{GMR,inh}$ is a parameter measuring the effect on bias of studying a mechanism-based inhibition. Variable $\mathbb{I}_{str_i}$ is an indicator function taking the value 0 if study i used a weak inhibitor, 1 if moderate inhibitor, and 2 if strong inhibitor; $\delta_{GMR,inh}$ is a parameter measuring the effect inhibitor strength on bias. This effect is linear: If $\mathbb{I}_{str_i} = 0$, $\beta'_{GMR,i}$ has a baseline value, if $\mathbb{I}_{str_i} = 1$, the value $\delta_{GMR,str}$ gets added to $\beta'_{GMR,i}$, if $\mathbb{I}_{str_i} = 2$, the value $2 \times \delta_{GMR,str}$ gets added to $\beta'_{GMR,i}$.

The true variance in log-space of $GMR_{i,obs}$, $\sigma^2_{GMR,i}$, is again assumed to be equal to the sum of between-study variance, $\sigma^2_{stu,i}$ (which is assumed to depends on the CYP enzyme mediating the studied interaction and on the type of interaction occurring), and a sampling variance equal to the true between-subject variance $\sigma^2_{sub,i}$, scaled by $N_{sub,i}$, the number of subject in study i:

$$\sigma^2_{GMR,i} = \sigma^2_{stu,i} + \frac{\sigma^2_{sub,i}}{N_{sub,i}} \quad (4)$$

Between-study variance, $\sigma^2_{stu,i}$ is computed as:

$$\sigma^2_{stu,i} = \nu_{stu,CYP_i} \times \exp\left(\mathbb{I}_{inh_i} \times \nu_{stu,inh} + \mathbb{I}_{str_i} \times \nu_{GMR,str}\right) \quad (5)$$

Parameter ν_{stu,CYP_i} is a CYP-specific between-study variance affecting simulations of the CYP examined in study i; $\mathbb{I}_{inh_i}$ and $\mathbb{I}_{str_i}$ are the same indicator functions as above; $\nu_{GMR,inh}$ is a parameter measuring the effect on between-study variance of studying a mechanism-based inhibition. Parameter $\nu_{GMR,inh}$ measures the linear effect (within the exponential!) of inhibitor strength on between-study variance.

The parameters $\delta_{GMR,str}$ and $\nu_{stu,str}$ are assigned normal priors:

$$\delta_{GMR,str} \sim \mathcal{N}(0,1) \quad (6)$$

$$\nu_{stu,str} \sim \mathcal{N}(0,1) \quad (7)$$

Between-study variances, $\sigma^2_{stu,i}$, may include a component due to Simcyp's imprecision in predictions, but this component is not separately estimable.

The true between-subject variance in log-space for study i was assumed to be equal to the Simcyp-predicted between-subject variance (in log-space), corrected for potential variance bias, $\beta_{BSV,i}$, as EMA's implementation of the previous model:

$$\sigma^2_{sub,i} = \sigma^2_{sub,i,pred} \times \exp\left(\beta'_{BSV,i}\right) \quad (8)$$

Here also, a hierarchical specification for $\beta'_{BSV,i}$ allows for information sharing in BSV predictions bias estimates:

$$\beta'_{BSV,i} \sim \mathcal{N}\left(\mu'_{\beta BSV}, {\sigma'_{\beta BSV}}^2\right) \quad (9)$$

As in the previous model, with EMA reparameterization, the observed between-subject variance (in log-space) for study i is part of the data and assumed to be distributed according to the following gamma distribution, specified through shape and rate parameters k_i and τ_i respectively:

$$\sigma^2_{sub,i,obs} = \left[log\left(GSD_{i,obs}\right)\right]^2 \sim G\left(k_i, \tau_i\right) \quad (10)$$

$$k_i = N_{sub,i}/2 \quad (11)$$

$$\tau_i = k_i/\sigma^2_{sub,i} \quad (12)$$

The priors for the parameters of the log GMR prediction bias, δ_{GMR,CYP_i} and $\delta_{GMR,inh}$, were both normal:

$$\delta_{GMR,CYP_i} \sim \mathcal{N}(0,1) \quad (13)$$

$$\delta_{GMR,inh} \sim \mathcal{N}(0,1) \quad (14)$$

The priors for the parameters of the BSV prediction bias, $\mu'_{\beta BSV}$ and $\sigma'_{\beta BSV}$, were again normal and truncated Cauchy:

$$\mu'_{\beta BSV} \sim \mathcal{N}(0,1) \quad (15)$$

$$\sigma'_{\beta BSV} \sim C(0,1)\ [0,\infty[\quad (16)$$

The priors for the parameters of between-study variances, $\sigma^2_{stu,i}$, ν_{stu,CYP_i} and $\nu_{stu,inh}$, were respectively assigned a Cauchy prior, truncated to positive values and a standard normal:

$$\nu_{stu,CYP_i} \sim C(0,1)\ [0,\infty[\quad (17)$$

$$\nu_{stu,inh} \sim \mathcal{N}(0,1) \quad (18)$$

Bias and imprecision model E

Model E just uses a nonlinear (quadratic) version of the effect of inhibitor strength.

Bias for study *i* is given by:

$$\beta'_{GMR,i} = \delta_{GMR,CYP_i} + \mathbb{I}_{inh_i} \times \delta_{GMR,inh} + \mathbb{I}_{str_i} \times \delta_{GMR,str,a} + \mathbb{I}^2_{str_i} \times \delta_{GMR,str,b} \quad (19)$$

The effect inhibitor strength on bias is null if $\mathbb{I}_{str_i} = 0$; otherwise quadratic in $\mathbb{I}_{str_i}$, with parameters $\delta_{GMR,str,a}$ and $\delta_{GMR,str,b}$.

Between-study variance, $\sigma^2_{stu,i}$ is computed as:

$$\sigma^2_{stu,i} = \nu_{stu,CYP_i} \times \exp\left(\mathbb{I}_{inh_i} \times \nu_{stu,inh} + \mathbb{I}_{str_i} \times \nu_{GMR,str,a} + \mathbb{I}^2_{str_i} \times \nu_{GMR,str,b}\right) \quad (20)$$

The effect inhibitor strength on between-study variance is null if $\mathbb{I}_{str_i} = 0$; otherwise quadratic in $\mathbb{I}_{str_i}$, with parameters $\nu_{GMR,str,a}$ and $\nu_{GMR,str,b}$.

The parameters $\delta_{GMR,str,a}$, $\delta_{GMR,str,b}$, $\nu_{stu,str,a}$, and $\nu_{stu,str,b}$ are assigned normal priors:

$$\delta_{GMR,str,a} \sim \mathcal{N}(0,1) \quad (21)$$

$$\delta_{GMR,str,b} \sim \mathcal{N}(0,1) \quad (22)$$

$$\nu_{stu,str,a} \sim \mathcal{N}(0,1) \quad (23)$$

$$\nu_{stu,str,b} \sim \mathcal{N}(0,1) \quad (24)$$

Methods and software

As previously, inference for models D and E was performed using Hamiltonian Markov chain Monte Carlo (HMCMC) simulations with the *R* package *RStan* and with all post-processing and plotting done in *R*.

Results

Convergence of MCMC and comparison of goodness of fit for models B, D, and E

HMCMC sampling converged well for all models. Gelman and Rubin $\hat{R}$ diagnostic values (which should be close to 1 at convergence) were at most 1.006 for all samples variables of the three models.

To assess goodness of fit, Figure 2 compares the distributions of published observed geometric ratios (top row) and between-subject variability in individual ratios (bottom row) to distributions of pseudo-data simulated using the parameters' joint posterior distribution of models B, D, and E. The leftmost panel of the Figure also shows the distribution of the published observations and raw Simcyp® Simulator predictions, for reference. This illustrates the problem mentioned in the presentation: the fits seem reasonable, but those plots do not help in selecting the best model, they look absolutely the same.



Figure 2: Fit of observed (dark lines) geometric ratios (top row) and between-subject variability in individual ratios (bottom row) with raw Simcyp predictions (left column) or posterior bias model predictions simulated using the parameters joint posterior distribution for models B, D, and E (three rightmost columns, respectively).

This lack of discriminating power is not even remedied if we stratify by strength of inhibitor (one of the EMA requests) (Figure 3).



Figure 3: Fit of model B, D, and E posterior bias predictions (columns 1, 2, and 3, respectively) in the case of weak, moderate, and strong inhibitors (top, middle, bottom panes, respectively) to observed (dark lines) geometric ratios (top row of each pane) and between-subject variability in individual ratios (bottom row of each pane).

The only difference is that models D and E can make specific predictions for weak, moderate and strong inhibitors, while model B cannot.

We should recall that those models are quite peculiar. They correct for biases. If we were *not* grouping or hierarchising by CYP, *etc*., they would *perfectly* correct all biases and the fit of predictions to data would be perfect. This is because there would be as many parameters as data points (studies in this case). Grouping introduces some mismatch, but it is quite subtle. Overall, the analysis tells us that the original mismatch between data and Simcyp predictions is of the order of percents. After correction for bias, we are down to small fractions of percents for mismatch and basically all models look quite good (and probably are).

So, maybe statisticians in this case should yield to decision makers: if decision makers want to make specific predictions for weak, moderate and strong inhibitors, models D and E are better. However, decision makers like to be reassured about the quality of the models founding the decisions: are models D and E equally good? If not, does it matter for precisions and decisions, and which one is best? So, we focus now on model checking and model comparison.

Models D and E maximum likelihood and maximum posterior probability comparisons

Bayesian inference is based on statistical modelling and aims to obtain the joint posterior probability distribution of all estimands. A key quantity here is the value of the maximum posterior density (MPD) of the parameter sample. MPD depends on both data likelihood and prior probability. We do not rely on informative priors here and the priors vary across models (but not the data), so we could also examine the maximum likelihood value (MLV) of the parameter sample. Both MPD and MLV can be taken as estimates of model quality. Those are given in Table 1. Unfortunately, they do not really help choosing between models D and E. Model D seems to fare better as far as MLV is concerned (but not by much, given the number of the parameters) and the MPD gives the inverse ranking… We turn next into looking at how reasonable the parameter estimates are.

Table 1: Maximum likelihood value (MLV) and maximum posterior density (MPD) obtained for the three models considered.

Criterion	Model B	Model D	Model E
MLV	37.89	42.43	34.99
MPD	107.38	101.72	104.34

Parameter estimates and respect of assumptions

We can examine the parameter estimates obtained (as samples from their joint posterior distribution). Table 2 gives the means and SDs of the parameters' marginal posterior

distributions for models B, D, and E. The results show an overall stability of the bias estimates. The mean GMR biases are inflated in models D and E, compared to model B, but remember that a new covariate (strength) is considered in the last two models, so the raw mean GMR biases are in fact reflective of the weak inhibitor class only. The biases for the moderate and strong inhibitor classes are different. So, in essence, this does not help us either in choosing between models D and E.

It might be better to examine plots of the covariate-corrected bias estimates. Figure 4 shows the estimates of GMR bias according to model B and model D, for the various cases of covariate values. Figure 5 does the same for models B and E. Figure 6 and Figure 7 are for between-study variance (the upper bound for Simcyp imprecision). Basically, there are no differences between models D and E here again (and even with model B). Note that they do not suffer from the pathologies induced in model C by the low number of studies. So, at least, models D and E are better than model C.

In those conditions, we are left with examining the shape of the covariate model itself (Figure 8 and Figure 9). Those make it clear that the effect of inhibitor strength is small on GMR bias and that there may be a small effect on between-study variability (it being slightly higher in the base of strong inhibitors). Retrospectively, the effects seen with model C were likely to be due to the reduced data set.

Table 2: Statistical summaries of the posterior distributions of the main parameters for models B, D, and E. The latter two have inhibitor strength as a covariate, model D in a linear fashion, model E in a quadratic fashion.

Parameter	**Model B**		**Model D**		**Model E**	
	Mean	SD	Mean	SD	Mean	SD
Mean GMR biases*						
CYP1A2	-0.0709	0.0314	-0.0945	0.0422	-0.0888	0.0422
CYP2C19	-0.1019	0.1011	-0.1371	0.1042	-0.1397	0.1074
CYP2C8	-0.0332	0.0628	-0.0512	0.0661	-0.0520	0.0628
CYP2C9	-0.0600	0.0393	-0.0817	0.0443	-0.0605	0.0587
CYP2D6	-0.1661	0.0539	-0.1937	0.0569	-0.1914	0.0577
CYP3A4	-0.0083	0.0309	-0.0345	0.0388	-0.0284	0.0409
Between-study variances*						
CYP1A2	0.0213	0.0102	0.0247	0.0123	0.0236	0.0118
CYP2C19	0.1471	0.0657	0.1294	0.0596	0.1244	0.0596
CYP2C8	0.0306	0.0245	0.0277	0.0219	0.0256	0.0204
CYP2C9	0.0140	0.0104	0.0139	0.0105	0.0145	0.0111
CYP2D6	0.0641	0.0266	0.0571	0.0243	0.0546	0.0230
CYP3A4	0.0327	0.0093	0.0294	0.0087	0.0284	0.0085
Effects of inhibition type						
on bias	0.0389	0.0448	0.0280	0.0463	0.0346	0.0480
on between-study variance	0.2798	0.2195	0.2732	0.2182	0.2663	0.2119
Effects of inhibitor strength						
linear on bias	-†	-†	0.0231	0.0223	-0.0267	0.0974
quadratic on bias	-†	-†	-†	-†	0.0242	0.0466
linear on between-study variance	-†	-†	0.0715	0.0669	0.0656	0.0609
quadratic on between-study variance	-†	-†	-†	-†	0.0276	0.0263
BSV‡ bias mean	1.3994	0.0950	1.4063	0.0933	1.4065	0.0946
BSV bias SD	1.2853	0.0720	1.2850	0.0735	1.2839	0.0755

* For model B those estimates are for the competitive inhibition (CI) case; For models D and E, they are for the case of CI and weak inhibitors.
‡ BSV: Between-subject variance.
† Not computed by the model.



Figure 4: Comparison of the posterior distributions of mean GMR biases by CYP for models B and D. X-axis labels: Model B: B1 for competitive inhibition case; B2 for MBI case; Model D: D1*x* for competitive inhibition; D2*x* for MBI; D*x*1 for a weak inhibitor; D*x*2 for a moderate inhibitor; D*x*3 for a strong inhibitor.



Figure 5: Comparison of the posterior distributions of mean GMR biases by CYP for models B and E. X-axis labels: Model B: B1 for competitive inhibition case; B2 for MBI case; Model E: E1*x* for competitive inhibition; E2*x* for MBI; E*x*1 for a weak inhibitor; E*x*2 for a moderate inhibitor; E*x*3 for a strong inhibitor.



Figure 6: Comparison of the posterior distributions of between-study variances by CYP for models B and D. X-axis labels: Model B: B1 for competitive inhibition case; B2 for MBI case; Model D: D1*x* for competitive inhibition; D2*x* for MBI; D*x*1 for a weak inhibitor; D*x*2 for a moderate inhibitor; D*x*3 for a strong inhibitor.



Figure 7: Comparison of the posterior distributions of between-study variances by CYP for models B and E. X-axis labels: Model B: B1 for competitive inhibition case; B2 for MBI case; Model E: E1*x* for competitive inhibition; E2*x* for MBI; E*x*1 for a weak inhibitor; E*x*2 for a moderate inhibitor; E*x*3 for a strong inhibitor.



Figure 8: Posterior draws of the strength covariate model for effect on GMR bias. Left pane: model D; right pane: model E. Basically, the effect of strength is identified to be small and quite uncertain.



Figure 9: Posterior draws of the strength covariate model for effect on between-study variance (an upper bound for Simcyp precision). Left pane: model D; right pane: model E. Overall, modest and quite uncertain effect of strength is identified. Model E may describe it a bit better.

Looking back at model B

- Replicate various goodness-of-fit graphs, results and posterior predictive visualizations for Model B. In addition, please submit posterior distributions of the estimated parameters stratified by CYP and type of inhibition for Model B only ("clean Model B" plots [i.e. without any comparisons to Models A and C] is considered useful for including in a potential opinion document).

Model B structure and methods

Those have been described in detail in the document "Answers to EMA Third List of Issues´.

Model B goodness of fit graphs

The goodness of fir (Figure 10) is quite reasonable and much improved compared to the raw data and predictions.



Figure 10: Fit of observed (dark lines) geometric ratios (top row) and between-subject variability in individual ratios (bottom row) with raw Simcyp predictions (left column) or posterior bias model predictions simulated using the parameters joint posterior distribution for models B (right column).

Model B parameter estimates

Table 2 gives the means and SDs of the parameters' marginal posterior distributions for model B. Figure 11 shows a comparison of mean GMR bias for model B, according to CYP and inhibition type. Figure 12 does the same for between-study variance estimates. Mean GMR biases range from about -17% (for CYP2D6) to -1% in the case of competitive inhibition. Those estimates are reasonably well-identified. In the case of MBI, model B estimates a 4% up-shift in those values; the range would therefore be from -13% to 4% approximately, with less bias overall.

Between-study variances (on the log-scale) range from 0.014 to 0.15; this corresponds to CVs ranging from 12% to 40% on the natural scale. Remember that this is an upper limit of Simcyp imprecision. Models B identifies an increase by a factor 1.3 for each CYP in the case of mechanism-based inhibition compared to competitive inhibition.

The mean bias in between-subject variability is about a factor 2 (that is, the ratio between observed and predicted between-subject CV is $\sqrt{\exp(1.40)} = 2.0$) according to model B, with a large variability across studies.

Overall, model B allows for pooling of information (this is obvious given its structure) and it yields estimates quite homogeneous and precise. This is particularly true in the case of between-subject variability bias, which should be better estimated in model B.

Table 3: Statistical summaries of the posterior distributions of the main parameters in model B.

Parameter	**Model B**	
	Mean	SD
Mean GMR biases*		
CYP1A2	-0.0709	0.0314
CYP2C19	-0.1019	0.1011
CYP2C8	-0.0332	0.0628
CYP2C9	-0.0600	0.0393
CYP2D6	-0.1661	0.0539
CYP3A4	-0.0083	0.0309
Between-study variances*		
CYP1A2	0.0213	0.0102
CYP2C19	0.1471	0.0657
CYP2C8	0.0306	0.0245
CYP2C9	0.0140	0.0104
CYP2D6	0.0641	0.0266
CYP3A4	0.0327	0.0093
Effects of inhibition type		
on bias	0.0389	0.0448
on between-study variance	0.2798	0.2195
BSV‡ bias mean	1.3994	0.0950
BSV bias SD	1.2853	0.0720

* Estimates for the competitive inhibition (CI) case.
‡ BSV: Between-subject variance.



Figure 11: Comparison of the posterior distributions of mean GMR biases by CYP for model B, in the case of competitive inhibition (left pane) and in the case of MBI (right pane).



Figure 12: Comparison of the posterior distributions of between-study variances by CYP for model B, in the case of competitive inhibition (left pane) and in the case of MBI (right pane)..

Model B posterior predictive plots

EMA proposed several predictive plots generated using the above model-based discrepancy analysis to help drug development. Such plots are presented below for model B.

Credibility interval vs predicted GMR

The above meta-analysis models can be used to understand how uncertainty affects future DDI predictions for CYP inhibition for regulatory decision-making.

Figure 13 displays 90% credibility intervals for GMRs (*i.e.*, fold-changes) according to model B, by CYP and by type of inhibition. Posterior samples of GMR bias and between-study variability were used to sample a log-normal distribution, as per the scripts provided by EMA. The results are quite consistent across covariate effects: predictions of GMRs for CYP2C19 have the highest uncertainty, followed by CYP2D6.



Figure 13: Model B-generated credibility intervals of predicted GMR values *vs*. those values for competitive inhibition (left pane) or mechanism-based inhibition (right pane). The x-axis values represent hypothetical GMR point estimates predicted by model B. The grey shaded are 90% credibility intervals of the prediction.

Predicted GMR for hypothetical CYP substrates

EMA suggested that the above displays could be extended to include information about therapeutic range. Figure 14 shows the results obtained for a hypothetical scenario in which:

- A CYP substrate is being developed for which there is an adequate PBPK model.
- The therapeutic index is known for the drug in question and was hypothetically set to 0.5 to 2-fold compared to the expected geometric mean exposure at the therapeutic dose.
- A hypothetical DDI is predicted using the Simcyp® Simulator following concomitant administration with CYP inhibitor. The type of inhibition is considered. Model B posteriors were used, as above, to estimate GMR uncertainty.



Figure 14: Model B predicted GMR following CYP inhibition for hypothetical CYP substrates in the case of competitive inhibition (left pane) or mechanism-based inhibition (right pane). The grey shaded area represents the therapeutic window. The red vertical line indicates its upper limit. The dashed vertical line indicates a predicted GMR without CYP inhibition. The blue dot represents the point estimate of the GMR predicted by the Simcyp® platform. The error bar gives the 90% credibility interval associated with the predicted GMR.

Probability of exceeding a given therapeutic index versus predicted GMR

The above displays were further processed to give the probability (estimated according to model B) of exceeding the therapeutic window when several hypothetical therapeutic windows are considered.



Figure 15: Probability, according to model B, of exceeding upper limit of the therapeutic index *vs*. predicted GMR for competitive inhibition (left pane) or mechanism-based inhibition (right pane). The predicted GMRs on the x-axis were predicted using Simcyp®. Results for hypothetical therapeutic index upper limits of 2-, 3-, 4- and 5-fold are displayed.

Maximum predicted GMR for less than 5% risk of exceeding a given therapeutic index

The above analysis can be expanded to determine the predicted GMR that would lead to a 5% probability of exceeding the upper limit of a therapeutic window. The threshold of 5% is a preliminary proposal of EMA.



Figure 16: Maximum predicted GMR for <5% risk of exceeding the two-fold upper limit of a therapeutic index for competitive inhibition (left pane) or mechanism-based inhibition (right pane). The x-axis shows the maximum GMR predicted by Simcyp® with <5% risk of exceeding the upper limit of therapeutic index given the uncertainty obtained from model B analysis.

Further goodness of fit graphs for models B and D

- Present GoF plots comparing the observed and predicted distributions of GMR bias and extra_var_of_ratio for Models B and D with some additional stratifications:
  - Stratify on strong, moderate and weak inhibition;
  - Generate plots where moderate and weak inhibition are in the same panel;
  - Stratify on inhibition type (competitive versus MBI).

Goodness of fit stratified by inhibitor strength

Figure 17 show goodness of fit stratified by inhibitor strength for model B and D. The fits are quite similar.

Goodness of fit by inhibitor strength, regrouping weak and moderate inhibitors

Figure 18 show goodness of fit for this stratification for model B and D. Here also, the fits are quite similar.

Goodness of fit stratified by type of inhibition

Figure 19 show goodness of fit for this stratification for model B and D. Here also, there is not appreciable difference in goodness of fit between the two models.



Figure 17: Fit of model B (top frame) and D (bottom frame) posterior bias predictions in the case of weak, moderate, and strong inhibitors (left, middle, right panes, respectively) to observed (dark lines) geometric ratios (top row of each pane) and between-subject variability in individual ratios (bottom row of each pane).



Figure 18: Fit of model B (top frame) and D (bottom frame) posterior bias predictions in the case of weak *or* moderate inhibitors (left column), and strong inhibitors (right column) to observed (dark lines) geometric ratios (top row of each pane) and between-subject variability in individual ratios (bottom row of each pane).



Figure 19: Fit of model B (top frame) and D (bottom frame) posterior bias predictions in the case of CI (left column) and MBI (right colum) to observed (dark lines) geometric ratios (top row of each pane) and between-subject variability in individual ratios (bottom row of each pane).

Conclusions

We have developed and compared two models taking inhibitor strength as a covariate effect (model D, linear, and model E, quadratic) to model B. It can be concluded that:

- The two models accounting for inhibitor strength (D and E) are identifiable and both point to a low effect of strength on bias and imprecision.
- The overall estimates of Simcyp bias and imprecision are rather low, similar and robust.
- Models D and E are more general than model B and could be prefered on that basis, even though they do not make much difference in the case of Simcyp. For simplicity and parcimony, model B could be preferred.

- Between model D and model E, model E is more flexible and may be preferred for future uses. It seems to perform slightly better for imprecision estimation in the case of Simcyp.
- Using different types of diagostics, like those used here to compare models D and E, are recommended to judge model fit and for model checking.

References

1. R Development Core Team. R: A Language and Environment for Statistical Computing, http://www.R-project.org. Vienna, Austria: R Foundation for Statistical Computing; 2013.

Appendix

Model D code ("EMA extended m203.stan")

```
// R-code for fitting Bayesian meta-regression model
// SAWP model in Stan, modified to include inhibitor strength with three
// classes ("m203.stan")

// ------------------
data {
  int<lower=0> N;
  int<lower=0> K;
  int<lower=0> L;
  int NSub[N];
  int CYP[N];
  int TIN[N]; // indicator of inhibition type;     1=CI; 2=MBI
  int SIN[N]; // indicator of inhibitor strength; 1=weak; 2=moderate; 3=strong
  vector[N] Obs_BSV_of_ratio;
  vector[N] Obs_ratio;
  vector[N] BSV_simcyp_pred;
  vector[N] simcyp_ratio;
}

// ------------------
parameters {
  vector[K]          ratio_bias_CYP;
  real               ratio_bias_TIN;
  real               ratio_bias_SIN;
  real               mean_BSV_bias;
  real<lower=0>      log_sd_BSV_bias;
  vector<lower=0>[K] extra_var_CYP;
  real<lower=0>      extra_var_TIN;
  real<lower=0>      extra_var_SIN;
  vector[N]          BSV_bias_std;
}

// ------------------
transformed parameters {
  vector[N] mean_bias;
  vector[N] log_corrected_ratio;
  vector[N] BSV_bias;
  vector[N] true_BSV_of_ratio;
  vector[N] shape;
  vector[N] scale;
  vector[N] rate;
```

```
  vector[N] extra_var_of_ratio;
  vector[N] total_var_of_ratio;
  vector[N] total_SD_of_ratio;

  for (n in 1:N){
    // uses indicator variables.
    mean_bias[n] = ratio_bias_CYP[CYP[n]] + (TIN[n] - 1) * ratio_bias_TIN +
                                            (SIN[n] - 1) * ratio_bias_SIN;

    log_corrected_ratio[n] = log(simcyp_ratio[n]) + mean_bias[n];

    BSV_bias[n] = mean_BSV_bias + BSV_bias_std[n] * log_sd_BSV_bias;

    true_BSV_of_ratio[n] = exp(log(BSV_simcyp_pred[n]) + BSV_bias[n]);

    shape[n] = NSub[n]/2; // alpha in STAN
    scale[n] = true_BSV_of_ratio[n]/shape[n];
    rate[n]  = 1/scale[n]; // beta in STAN

    // uses indicator variables.
    extra_var_of_ratio[n] = extra_var_CYP[CYP[n]] *
                            exp((TIN[n] - 1) * extra_var_TIN +
                            (SIN[n] - 1) * extra_var_SIN);

    total_var_of_ratio[n] = true_BSV_of_ratio[n] / NSub[n] +
                            extra_var_of_ratio[n];

    total_SD_of_ratio[n] = sqrt(total_var_of_ratio[n]);
  }
}

// ------------------
model {
  // adaptive priors (describing the random effects distributions)
  BSV_bias_std ~ normal(0, 1);

  // hyper-priors (describing the priors for the parameters for the
  // random effects distributions)
  mean_BSV_bias   ~ normal(0, 1); // log scale; we use U(0,50)
  log_sd_BSV_bias ~ cauchy(0, 1); // log scale; we used half truncated normal

  // regular priors
  ratio_bias_CYP ~ normal(0, 1);
  ratio_bias_TIN ~ normal(0, 1);
  ratio_bias_SIN ~ normal(0, 1);

  extra_var_CYP  ~ cauchy(0, 1);
  extra_var_TIN  ~ normal(0, 1);
  extra_var_SIN  ~ normal(0, 1);

  // likelihood
  Obs_BSV_of_ratio ~ gamma(shape, rate);
  Obs_ratio        ~ lognormal(log_corrected_ratio, total_SD_of_ratio);
}

// ------------------
generated quantities {

  array[N] real predBSV = gamma_rng(shape, rate);
  array[N] real predR   = lognormal_rng(log_corrected_ratio,
                                        total_SD_of_ratio);

//  array[N] real predBSV2 = gamma_rng(shape*1000, 1000/scale);
//  array[N] real predR2   = lognormal_rng(log_corrected_ratio,
//                                         sqrt(extra_var_of_ratio));
//  vector[K] sim_log_corrected_ratio = log(sim_ratio) + mean_bias;
//  array[K] real simR   = lognormal_rng(sim_log_corrected_ratio,
//                                       sqrt(extra_var_of_ratio));
```

```
  vector[N] log_lik;
  for (n in 1:N) {
    log_lik[n] = gamma_lpdf(Obs_BSV_of_ratio[n] | shape[n], rate[n]) +
                 lognormal_lpdf(Obs_ratio[n] | log_corrected_ratio[n],
                                               total_SD_of_ratio[n]);
  }
}

// End.
```

Model E code ("EMA extended m204.stan")

```
// R-code for fitting Bayesian meta-regression model
// SAWP model in Stan, modified to include inhibitor strength with three
// classes and nonlinear effect of it ("m204.stan")

// ------------------
data {
  int<lower=0> N;
  int<lower=0> K;
  int<lower=0> L;
  int NSub[N];
  int CYP[N];
  int TIN[N]; // indicator of inhibition type;    1=CI; 2=MBI
  int SIN[N]; // indicator of inhibitor strength; 1=weak; 2=moderate; 3=strong
  vector[N] Obs_BSV_of_ratio;
  vector[N] Obs_ratio;
  vector[N] BSV_simcyp_pred;
  vector[N] simcyp_ratio;
}

// ------------------
parameters {
  vector[K]          ratio_bias_CYP;
  real               ratio_bias_TIN;
  real               ratio_bias_SIN_a;
  real               ratio_bias_SIN_b;
  real               mean_BSV_bias;
  real<lower=0>      log_sd_BSV_bias;
  vector<lower=0>[K] extra_var_CYP;
  real<lower=0>      extra_var_TIN;
  real<lower=0>      extra_var_SIN_a;
  real<lower=0>      extra_var_SIN_b;
  vector[N]          BSV_bias_std;
}

// ------------------
transformed parameters {
  vector[N] mean_bias;
  vector[N] log_corrected_ratio;
  vector[N] BSV_bias;
  vector[N] true_BSV_of_ratio;
  vector[N] shape;
  vector[N] scale;
  vector[N] rate;
  vector[N] extra_var_of_ratio;
  vector[N] total_var_of_ratio;
  vector[N] total_SD_of_ratio;

  for (n in 1:N){
    // uses indicator variables.
    mean_bias[n] = ratio_bias_CYP[CYP[n]] + (TIN[n] - 1) * ratio_bias_TIN +
                                            (SIN[n] - 1) * ratio_bias_SIN_b +
                                            (SIN[n] - 1)^2 * ratio_bias_SIN_a;

    log_corrected_ratio[n] = log(simcyp_ratio[n]) + mean_bias[n];
```

```
    BSV_bias[n] = mean_BSV_bias + BSV_bias_std[n] * log_sd_BSV_bias;

    true_BSV_of_ratio[n] = exp(log(BSV_simcyp_pred[n]) + BSV_bias[n]);

    shape[n] = NSub[n]/2; // alpha in STAN
    scale[n] = true_BSV_of_ratio[n]/shape[n];
    rate[n]  = 1/scale[n]; // beta in STAN

    // uses indicator variables.
    extra_var_of_ratio[n] = extra_var_CYP[CYP[n]] *
                            exp((TIN[n] - 1) * extra_var_TIN +
                            (SIN[n] - 1) * extra_var_SIN_b +
                            (SIN[n] - 1)^2 * extra_var_SIN_a);

    total_var_of_ratio[n] = true_BSV_of_ratio[n] / NSub[n] +
                            extra_var_of_ratio[n];

    total_SD_of_ratio[n] = sqrt(total_var_of_ratio[n]);
  }
}

// -------------------
model {
  // adaptive priors (describing the random effects distributions)
  BSV_bias_std ~ normal(0, 1);

  // hyper-priors (describing the priors for the parameters for the
  // random effects distributions)
  mean_BSV_bias   ~ normal(0, 1); // log scale; we use U(0,50)
  log_sd_BSV_bias ~ cauchy(0, 1); // log scale; we used half truncated normal

  // regular priors
  ratio_bias_CYP   ~ normal(0, 1);
  ratio_bias_TIN   ~ normal(0, 1);
  ratio_bias_SIN_a ~ normal(0, 1);
  ratio_bias_SIN_b ~ normal(0, 1);

  extra_var_CYP    ~ cauchy(0, 1);
  extra_var_TIN    ~ normal(0, 1);
  extra_var_SIN_a  ~ normal(0, 1);
  extra_var_SIN_b  ~ normal(0, 1);

  // likelihood
  Obs_BSV_of_ratio ~ gamma(shape, rate);
  Obs_ratio        ~ lognormal(log_corrected_ratio, total_SD_of_ratio);
}

// -------------------
generated quantities {

  array[N] real predBSV = gamma_rng(shape, rate);
  array[N] real predR   = lognormal_rng(log_corrected_ratio,
                                        total_SD_of_ratio);

  vector[N] log_lik;
  for (n in 1:N) {
    log_lik[n] = gamma_lpdf(Obs_BSV_of_ratio[n] | shape[n], rate[n]) +
                 lognormal_lpdf(Obs_ratio[n] | log_corrected_ratio[n],
                                               total_SD_of_ratio[n]);
  }
}

// End.
```